5f. Relations for normalized Hermite functions

See Table 2.4

Normalized Hermite functions for symbolic computation

(**Hermn** defined in 5e denotes the explicit function, here **hermn** is defined for symbolic manipulation)

In[ ]:=
```
Clear[hermn, xi, hermsub, dif]
dif[aa_ + bb_] := dif[aa] + dif[bb]
dif[ ff_ aa_] := ff dif[aa] /; FreeQ[ff, hermn]
hermsub[x_] := Module[{xx}, xx = Expand[x /. hermrel];
   xx //. hermrel // Simplify]
hermrel = {dif[hermn[ell_, k_]] :→
      Sqrt[4 Pi Abs[ell]] ( Sqrt[k / 2] hermn[ell, k - 1] - Sqrt[(k + 1) / 2] hermn[ell, k + 1]),
    xi hermn[ell_, k_] :→ (4 Pi Abs[ell]) ^ (- 1 / 2)
      ( Sqrt[k / 2] hermn[ell, k - 1] + Sqrt[(k + 1) / 2] hermn[ell, k + 1]),
    xi ^ n_ hermn[ell_, k_] :→ xi ^ (n - 1) (4 Pi Abs[ell]) ^ (- 1 / 2)
      ( Sqrt[k / 2] hermn[ell, k - 1] + Sqrt[(k + 1) / 2] hermn[ell, k + 1])};
```

Check of relations for some values

In[ ]:=
```
Clear[rel]
rel[k_] = dif[hermn[ell, k]] // hermsub
Do[Print[k, "   ",
   D[Hermn[ell, k, xi], xi] == rel[k] /. hermn[el_, m_] :→ Hermn[el, m, xi] // Simplify], {k, 0, 5}]
```

Out[ ]= $\sqrt{2\pi}\ \sqrt{\text{Abs[ell]}}\left(\sqrt{k}\ \text{hermn[ell, -1 + k]} - \sqrt{1+k}\ \text{hermn[ell, 1 + k]}\right)$

```
0   True
1   True
2   True
3   True
4   True
5   True
```

In[ ]:=
```
Clear[rel]
rel[k_] = xi hermn[ell, k] // hermsub
Do[Print[k, "   ",
   xi Hermn[ell, k, xi] == rel[k] /. hermn[el_, m_] :→ Hermn[el, m, xi] // Simplify], {k, 0, 5}]
```

Out[ ]= $\frac{\sqrt{k}\ \text{hermn[ell, -1 + k]} + \sqrt{1+k}\ \text{hermn[ell, 1 + k]}}{2\ \sqrt{2\pi}\ \sqrt{\text{Abs[ell]}}}$

```
0   True
1   True
2   True
3   True
4   True
5   True
```

In[ ]:=
```
Clear[rel]
rel[k_] = xi ^ 2 hermn[ell, k] // hermsub
Do[Print[k, "   ",
  xi ^ 2 Hermn[ell, k, xi] == rel[k] /. hermn[el_, m_] :> Hermn[el, m, xi] // Simplify], {k, 0, 5}]
```

Out[ ]=

$$\frac{1}{8\,\pi\,\mathrm{Abs}[\mathrm{ell}]}\left(\sqrt{-1+k}\;\sqrt{k}\;\mathrm{hermn}[\mathrm{ell},\,-2+k]+\mathrm{hermn}[\mathrm{ell},\,k]+2\,k\,\mathrm{hermn}[\mathrm{ell},\,k]+\sqrt{1+k}\;\sqrt{2+k}\;\mathrm{hermn}[\mathrm{ell},\,2+k]\right)$$

```
0   True
1   True
2   True
3   True
4   True
5   True
```

This gives confidence that we use the right relations.